


Monkey Launcher

Getting Started

Start a new project and delete the cat sprite. Create 6 new sprites. You will need to draw the cannon and the base, but you can use the library for the monkey, palm tree, and bananas.




For the Cannon, draw a rectangle.



In order to get the cannon and the monkey to line up correctly, you will need to turn them on their sides so they are pointing to the right (see the picture at the top of the page). Also, **don't forget to center them** - make the center of the monkey the bottom of its feet, and the center of the cannon the very left edge (after it has been turned on its side).


To make the base, draw a square and shave off the top two corners.



When we put the cannon together, it will look something like this.



Make new variables for all sprites called **LaunchSpeed**, **FallSpeed**, **Gravity**, **Launches**, and **Bananas**.



Monkey Launcher

Monkey Code

```
when I receive Setup  
go to Cannon  
go to front  
go back 2 layers  
set size to 40 %  
show
```

```
when I receive Action  
point in direction direction of Cannon
```

```
when I receive LaunchMonkey  
set FallSpeed to 0  
repeat until touching edge ? or touching Palmtree ?  
  move LaunchSpeed steps  
  change y by FallSpeed  
  change FallSpeed by Gravity  
go to Cannon  
set FallSpeed to 0
```

Cannon Code

```
when I receive Setup  
go to Base  
point in direction 45  
set size to 50 %  
go to front  
go back 1 layers  
show
```

```
define CheckAngle  
if direction < 5 then  
  point in direction 5  
  set LaunchSpeed to 6  
if direction > 89 then  
  point in direction 89  
  set LaunchSpeed to 14.5
```

```
when I receive Action  
Aim
```

```
define Aim  
CheckAngle  
if key down arrow pressed? then  
  turn 1 degrees  
  if LaunchSpeed > 1 then  
    change LaunchSpeed by .1  
if key up arrow pressed? then  
  turn 1 degrees  
  if LaunchSpeed < 20 then  
    change LaunchSpeed by -.1
```

Base Code:

```
when I receive Setup  
go to x: -208 y: -152  
go to front  
set size to 50 %  
show
```

Monkey Launcher


GameLoop Code:



```
when green flag clicked
broadcast Setup and wait
repeat until Bananas = 0
  broadcast Action and wait
broadcast GameOver
```




```
when I receive Setup
set LaunchSpeed to 13
set FallSpeed to 0
set Gravity to -2
set Launches to 0
```



```
when space key pressed
broadcast LaunchMonkey
change Launches by 1
```

Palm Tree Code:




```
when I receive Setup
go to front
set size to 60 %
go to x: 62 y: -108
show
```

Bananas Code:



```
when I receive Setup
hide
set Bananas to 5
repeat Bananas
  create clone of myself
```



```
when I start as a clone
go to x: pick random -110 to 230 y: pick random -160 to 60
set size to pick random 50 to 100 %
set color effect to pick random -10 to 20
show
wait until touching Monkey2 ?
change Bananas by -1
delete this clone
```

Your Challenge:

Set up the code that happens when “GameOver” is broadcast. How do you want the game to end?